
Subject: Re: Reading a set of data with string type entries

Posted by [AlbertoPaolo.Meroni](#) on Fri, 30 Nov 2007 07:32:40 GMT

[View Forum Message](#) <> [Reply to Message](#)

On 30 Nov, 08:07, izim...@gmail.com wrote:

```
> IDL> s = strjoin(['info', string(randomu(seed,10))], ' ')
> IDL> vals = float((strsplit(s, ' ', /regex, /extract))[1:*])
>
> For large files I sometimes read the whole thing into a bytarr and
> split on line breaks
> something like
>
> openr, lun, 'big.txt', /get_lun
> fs = fstat(lun)
> buf = bytarr(fs.size)
> readu, lun, buf
> free_lun, lun
> buf = strsplit(string(buf), string([13b,10b]), /extract) ; split on CR
> +LF
```

I suggest to use the following code:

```
; -----
FUNCTION ReadUnformattedAscii, FileName
  COMPILE_OPT IDL2
  IF ~FILE_TEST(FileName) THEN RETURN, 0
; # of line
  nLines = FILE_LINES(FileName)

  OPENR, Lun, FileName, /Get_Lun
; Read First Line to establish # of Values inside the record
  dummy = ""
  READF, Lun, Dummy
  POINT_LUN, Lun, 0
  Dummy = STRCOMPRESS(dummy)
  void = STRSPLIT(dummy, Count=nFields)
  nValues = NFields-1
; Define my output structure
  str_out = {Text: "", $
             Values: FLTARR(nValues)}
  str_out = REPLICATE(str_out, nLines)
; Define Tmp Vars
  TmpArr = FLTARR(nValues)
  TmpStr = ""
  FOR j = 0, nlines-1 DO BEGIN
    READF, Lun, Dummy
    dummy = STRCOMPRESS(dummy)
    IText = STRPOS(dummy, ' ')
```

```
    rForm = '(A'+STRCOMPRESS(STRING(IText))+','+ $
        STRCOMPRESS(STRING(nValues))+'F0)'
    READS,dummy,TmpStr,TmpArr,FORMAT=rForm
    str_out[j].Text = TmpStr
    Str_out[j].Values = TmpArr
ENDFOR
FREE_Lun,Lun
RETURN,str_out
END
; -----
```
